

G08BAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G08BAF performs Mood's and David's tests for dispersion differences between two independent samples of possibly unequal size.

2 Specification

```
SUBROUTINE G08BAF(X, N, N1, R, ITEST, W, V, PW, PV, IFAIL)
  INTEGER          N, N1, ITEST, IFAIL
  real            X(N), R(N), W, V, PW, PV
```

3 Description

Mood's and David's tests investigate the difference between the dispersions of two independent samples of sizes n_1 and n_2 , denoted by:

$$x_1, x_2, \dots, x_{n_1}$$

and

$$x_{n_1+1}, x_{n_1+2}, \dots, x_n, \quad n = n_1 + n_2.$$

The hypothesis under test, H_0 , often called the null hypothesis, is that the dispersion difference is zero, and this is to be tested against a one- or two-sided alternative hypothesis H_1 (see below).

Both tests are based on the rankings of the sample members within the pooled sample formed by combining both samples. If there is some difference in dispersion, more of the extreme ranks will tend to be found in one sample than in the other.

Let the rank of x_i be denoted by r_i , for $i = 1, 2, \dots, n$.

(a) Mood's test.

The test statistic $W = \sum_{i=1}^{n_1} (r_i - \frac{n+1}{2})^2$ is found.

W is the sum of squared deviations from the average rank in the pooled sample. For large n , W approaches normality, and so an approximation, p_w , to the probability of observing W not greater than the computed value, may be found.

G08BAF returns W and p_w if Mood's test is selected.

(b) David's test.

The disadvantage of Mood's test is that it assumes that the means of the two samples are equal. If this assumption is unjustified a high value of W could merely reflect the difference in means. David's test reduces this effect by using the variance of the ranks of the first sample about their mean rank, rather than the overall mean rank.

The test statistic for David's test is

$$V = \frac{1}{n_1 - 1} \sum_{i=1}^{n_1} (r_i - \bar{r})^2$$

where

$$\bar{r} = \frac{\sum_{i=1}^{n_1} r_i}{n_1}.$$

For large n , V approaches normality, enabling an approximate probability p_v to be computed, similarly to p_w .

G08BAF returns V and p_v if David's test is selected.

Hypothesis testing:

Suppose that a significance test of a chosen size α is to be performed (i.e., α is the probability of rejecting H_0 when H_0 is true; typically α is a small quantity such as 0.05 or 0.01).

The returned value p ($= p_v$ or p_w) can be used to perform a significance test, against various alternative hypotheses H_1 , as follows.

- (i) H_1 : dispersions are unequal. H_0 is rejected if $2 \times \min(p, 1 - p) < \alpha$.
- (ii) H_1 : dispersion of sample 1 $>$ dispersion of sample 2. H_0 is rejected if $1 - p < \alpha$.
- (iii) H_1 : dispersion of sample 2 $>$ dispersion of sample 1. H_0 is rejected if $p < \alpha$.

4 References

- [1] Cooper B E (1975) *Statistics for Experimentalists* Pergamon Press

5 Parameters

- 1:** X(N) — *real* array *Input*
On entry: the first n_1 elements of X must be set to the data values in the first sample, and the next n_2 ($= N - n_1$) elements to the data values in the second sample.
- 2:** N — INTEGER *Input*
On entry: the total of the two sample sizes, n ($= n_1 + n_2$).
Constraint: $N > 2$.
- 3:** N1 — INTEGER *Input*
On entry: the size of the first sample, n_1 .
Constraint: $1 < N1 < N$.
- 4:** R(N) — *real* array *Output*
On exit: the ranks r_i , assigned to the data values x_i , for $i = 1, 2, \dots, n$.
- 5:** ITEST — INTEGER *Input*
On entry: the test(s) to be carried out, using the codes:
 0 – both Mood’s and David’s tests
 1 – David’s test only
 2 – Mood’s test only
Constraint: ITEST = 0, 1 or 2.
- 6:** W — *real* *Output*
On exit: Mood’s test statistic, W , if requested.
- 7:** V — *real* *Output*
On exit: David’s test statistic, V , if requested.
- 8:** PW — *real* *Output*
On exit: the lower tail probability, p_w , corresponding to the value of W , if Mood’s test was requested.
- 9:** PV — *real* *Output*
On exit: the lower tail probability, p_v , corresponding to the value of V , if David’s test was requested.

10: IFAIL — INTEGER*Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, $N \leq 2$.

IFAIL = 2

On entry, $N1 \leq 1$,
or $N1 \geq N$.

IFAIL = 3

On entry, $ITEST < 0$,
or $ITEST > 2$.

7 Accuracy

All computations are believed to be stable. The statistics V and W should be accurate enough for all practical uses.

8 Further Comments

The time taken by the routine is small, and increases with n .

9 Example

This example is taken from page 280 of Cooper [1]. The data consist of two samples of six observations each. Both Mood's and David's test statistics and significances are computed. Note that Mood's statistic is inflated owing to the difference in location of the two samples, the median ranks differing by a factor of two.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G08BAF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          N
      PARAMETER       (N=12)
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real            PV, PW, V, W
      INTEGER          I, IFAIL, ITEST, N1
*      .. Local Arrays ..
      real            WK(N), X(N)
```

```

*    .. External Subroutines ..
EXTERNAL          G08BAF
*    .. Executable Statements ..
WRITE (NOUT,*) 'G08BAF Example Program Results'
*    Skip heading in data file
READ (NIN,*)
READ (NIN,*) X
N1 = 6
WRITE (NOUT,*)
WRITE (NOUT,*) 'Mood''s test and David''s test'
WRITE (NOUT,*)
WRITE (NOUT,*) 'Data values'
WRITE (NOUT,*)
WRITE (NOUT,99999) '    Group 1  ', (X(I),I=1,N1)
WRITE (NOUT,*)
WRITE (NOUT,99999) '    Group 2  ', (X(I),I=N1+1,N)
ITEST = 0
IFAIL = 0

*
CALL G08BAF(X,N,N1,WK,ITEST,W,V,PW,PV,IFAIL)

*
WRITE (NOUT,99998) '    Mood''s measure = ', W,
+ '    Significance = ', PW
WRITE (NOUT,99998) '    David''s measure = ', V,
+ '    Significance = ', PV
STOP

*
99999 FORMAT (1X,A,8F4.0,/(13X,8F4.0))
99998 FORMAT (1X,A,F8.3,A,F8.4)
END

```

9.2 Program Data

G08BAF Example Program Data

```

6.0  9.0 12.0  4.0 10.0 11.0
8.0  1.0  3.0  7.0  2.0  5.0

```

9.3 Program Results

G08BAF Example Program Results

Mood's test and David's test

Data values

```

Group 1    6.  9. 12.  4. 10. 11.

Group 2    8.  1.  3.  7.  2.  5.
Mood's measure =   75.500  Significance =   0.5830
David's measure =   9.467  Significance =   0.1986

```